<soap:Body>
    <Pay xmlns="PayService.Service">
      <partnerID>4L4D052A16WKWZG0</partnerID>
      <payInfo>
        <PayExtID>188598989256</PayExtID>
        <PayDate>2017-05-04T04:32:09</PayDate>
        <SerID>500284</SerID>
        <SerComINN>8601999247</SerComINN>
        <PayType>Principal</PayType>
        <Sum>10000</Sum>

        <Attributes>
          <Account>861000689319</Account>
          <AdditionalAccountAttributes>
            <Param>
              <Name>ИНН регионального оператора</Name>
              <Value>8602166551</Value>
            </Param>
          </AdditionalAccountAttributes>
        </Attributes>

        <AdditionalPayInfo>
          <Param>
            <Name>Филиал</Name>
            <Value>9067</Value>
          </Param>
        </AdditionalPayInfo>

      </payInfo>
    </Pay>
  </soap:Body>

Текст модуля

```1c
    WSОпределение = Новый WSОпределения(НашВебСервис);
    WSСервис=WSОпределение.Сервисы[0];
    прокси = Новый WSПрокси(WSОпределение, WSСервис.URIПространстваИмен, WSСервис.Имя, WSСервис.ТочкиПодключения[0].Имя);

    ТипWSДопПараметра = Прокси.ФабрикаXDTO.Пакеты.Получить((WSСервис.URIПространстваИмен)).Получить("Param");

    WSДопПараметр = Прокси.ФабрикаXDTO.Создать(ТипWSДопПараметра);

    WSДопПараметр.Name = "ИНН регионального оператора";
    WSДопПараметр.Value = "8602166551";

    ТипWSПараметра1 = Прокси.ФабрикаXDTO.Пакеты.Получить((WSСервис.URIПространстваИмен)).Получить("PayAttributes");

    WSПараметр1 = Прокси.ФабрикаXDTO.Создать(ТипWSПараметра1);
    WSПараметр1.Account = Стр.ЛицевойСчет.Код;

    WSПараметр1.AdditionalAccountAttributes = WSДопПараметр;// выдает ошибку

    Несоответствие типов XDTO:
    Тип '{PayService.Service}Param' не соответствует типу '{PayService.Service}ArrayOfParam'
    Тип свойства не соответствует типу значения

    ТипWSПараметра = Прокси.ФабрикаXDTO.Пакеты.Получить((WSСервис.URIПространстваИмен)).Получить("PayInfo");
    PayExtID = "VGRKAPREM"+Объект.Номер+"_"+Стр.НомерСтроки;

    WSПараметр = Прокси.ФабрикаXDTO.Создать(ТипWSПараметра);
    WSПараметр.PayExtID = PayExtID;
    WSПараметр.PayDate = Объект.Дата;
    WSПараметр.SerID = "500284";//КодУслуги взнос на кап ремонт
    WSПараметр.SerComINN = СокрЛП(Объект.Организация.ИНН);

    Если Объект.Пеня Тогда
    WSПараметр.PayType = "Peny";//ВидУслуги "Peny"
    Иначе
    WSПараметр.PayType = "Principal";//ВидУслуги "Основной долг"
    КонецЕсли;

    WSПараметр.Sum = Стр.Сумма*100;
    WSПараметр.Attributes = WSПараметр1;

    СпрПареметры = Справочники.РасшПараметры.НайтиПоКоду("000000001");
    ПутьКПапке = СпрПареметры.ПутьКПапкеЛоговПоВыгрузке;

    Прокси.Pay(НашКодПартнера,WSПараметр,PayStatus,Error);
```

Как передать значения в Param?
У AdditionalAccountAttributes тип Массив Param

  WSПараметр1.AdditionalAccountAttributes = WSДопПараметр;// выдает ошибку

                Несоответствие типов XDTO:
                Тип '{PayService.Service}Param' не соответствует типу '{PayService.Service}ArrayOfParam'
                Тип свойства не соответствует типу значения

Пример:

        <Attributes>
          <Account>861000689319</Account>
          <AdditionalAccountAttributes>
[ <Param>
              <Name>ИНН регионального оператора</Name>
              <Value>8602166551</Value>
            </Param>
          </AdditionalAccountAttributes>
           </Attributes>
---
Из кода видно, что AdditionalAccountAttributes - это СписокXDTO с типом "ArrayOfParam", а его содержимое - ЭлементXDTO с типом "Param".

```1c
    ТипArrayOfParam = Прокси.ФабрикаXDTO.Тип(WSСервис.URIПространстваИмен, "ArrayOfParam");
    ТипParam = Прокси.ФабрикаXDTO.Тип(WSСервис.URIПространстваИмен, "Param");

    ArrayOfParam = Прокси.ФабрикаXDTO.Создать(ТипArrayOfParam);
    Param = Прокси.ФабрикаXDTO.Создать(ТипParam);
    ...
    ArrayOfParam.Добавить(Param );
    ...
    WSПараметр1.AdditionalAccountAttributes = ArrayOfParam;
```
---
Из кода видно, что AdditionalAccountAttributes - это СписокXDTO с типом "ArrayOfParam", а его содержимое - ЭлементXDTO с типом "Param".

```1c
    ТипArrayOfParam = Прокси.ФабрикаXDTO.Тип(WSСервис.URIПространстваИмен, "ArrayOfParam");
    ТипParam = Прокси.ФабрикаXDTO.Тип(WSСервис.URIПространстваИмен, "Param");

    ArrayOfParam = Прокси.ФабрикаXDTO.Создать(ТипArrayOfParam);
    Param = Прокси.ФабрикаXDTO.Создать(ТипParam);
    ...
    ArrayOfParam.Добавить(Param );
    ...
    WSПараметр1.AdditionalAccountAttributes = ArrayOfParam;
```
---
ArrayOfParam = Прокси.ФабрикаXDTO.Создать(ТипArrayOfParam);

Спасибо вчера вечером уже голова не тумкала)))

Там только вместо ArrayOfParam.Добавить(Param );

надо было написать ArrayOfParam.Param.Добавить(Param );